

補遺 4

一次元探索 f_c の最適化プログラムと計算例

補遺 4 一次元探索 f_c の最適化プログラムと計算例

手法で採用した貯留関数モデルは次式で表わされる。

$$s = k_1 q^{p_1} + k_2 \frac{d}{dt}(q^{p_2}) \quad \dots\dots\dots (1)$$

$$\frac{ds}{dt} = r - q \quad \dots\dots\dots (2)$$

$$p_1 = 0.6 \quad \dots\dots\dots (3)$$

$$p_2 = 0.4648 \quad \dots\dots\dots (4)$$

$$k_1 = 2.8235 f_c A^{0.24} \quad \dots\dots\dots (5)$$

$$k_2 = 0.2835 k_1^2 \bar{r}^{-0.2648} \quad \dots\dots\dots (6)$$

ここで、 s ：貯留高 (mm)、 q ：流出高 (mm/hr)、 r ：有効雨量 (mm/hr)、 t ：時間 (hr)、 k_1 、 k_2 、 p_1 、 p_2 ：モデルパラメータ、 n ：等価粗度、 i ：斜面勾配、 A ：流域面積 (km²)。

(5)式に示される f_c 値は 1～2 のオーダーで分布していることが知られている。 f_c 値さえ決定できれば(1)式中のパラメータはすべて決定でき、流出計算が可能となる。

ここでは既往洪水資料から分離された雨量（有効雨量）と流量（直接流出高）を用いて、当該洪水に合う f_c 値を探索する。探索の手法は一次元的に f_c 値を変化させ、それぞれの f_c 値において決定されたパラメータを用いて流出計算を行い、実測流出高と計算流出高の差の 2 乗和を最小としたときの f_c 値が最適 f_c 値として同定される。この評価関数は次式に示される。

$$\text{Min}_{f_c} J(f_c) = \sum_{j=1}^N (q_j^* - q_j(f_c))^2 \quad \dots\dots\dots (7)$$

ここで、 q_j^* ：実測(分離)流出高 (mm/hr)、 $q_j(f_c)$ ：ある f_c 値が与えられたときの計算流出高 (mm/hr)、 N ：流出高標本数。




図-4・1 実測・計算ハイドログラフ




図-4・2 評価関数の変化

一般には、図-4・1に示すように計算値と実測値は必ずしも合致しない。しかしながら、 f_c の値を変化させて(7)式の値を計算していくと、図-4・2のような関係が見出せることが経験的に知られている。(7)式で計算される J の値が最小となるとき（実測と計算のハイドログラフが最も良く合ったとき）、その時点に設定した f_c 値が当該洪水の最適 f_c 値であると判断できる。ここでは、 f_c 値を一次的に探索する手法として比較的簡単な方法と数学的に求める方法の 2 種類を紹介する。

f_c 探索手法（その1）の説明

考え方は非常に簡単である。（ここでは、有効降雨データが既知である。）

- ① 流域面積 A ，モデルパラメータ p_1 ， p_2 ，有効雨量系列 $R(I)$ ，直接流出高系列 $Q(I)$ を入力する。
- ② 平均有効雨量強度 \bar{r} を計算する。
- ③ f_c 値の初期値を設定する。（オーダーが1～2であったことに留意する）
- ④ f_c 値のきざみ幅 Δf_c を設定する。（大体， $\Delta f_c = 0.1$ 程度）
- ⑤ $f_c = f_c + \Delta f_c$
- ⑥ ⑤で得られた f_c 値をもとに，(3)，(4)，(5)，(6)式によりモデルパラメータを全部決定する。
- ⑦ 補遺3のプログラムを用いて流量を計算する。
- ⑧ $J = \sum_{j=1}^N (q_j^* - q_j(f_c))^2$ を計算し，出力する。
- ⑨ ⑤に戻る。

⑤～⑨の作業を何回か行くと，図-4・2に示すように， J の値が最小となる所が必ず1箇所だけ出てくる。その時点に設定された f_c 値が当該洪水の最適 f_c 値である。①～⑨の手順は次項のフロー図中に示すので参照されたい。





図-4・3 f_c のハイドログラフに及ぼす影響

Δf_c 値を小さく設定すれば求める f_c 値の精度も高くなる。しかしながら， Δf_c 値を小さく設定すれば探索回数が増加し，計算時間がかかりすぎるおそれがある。したがって，実用上 f_c 値の初期値を 0.2， Δf_c 値を 0.1 程度に設定し，探索回数を 30 回程度とればよい。図-4・3 はここで示す手法を用いて f_c 値を逐次更新していったときのハイドログラフの変化である。 f_c 値は等価粗度と斜面勾配の関数である。1つの流域に着目し，斜面勾配の変化を無視できるものとする，変化することが予想されるのは等価粗度のみとなる。 f_c 値（等価粗度）が減少すると，ピーク流出高が増加し，ピーク生起時刻が早まる。したがって f_c 値は都市化に代表される流出形態の変化を表現できる。次に， f_c 探索手法（その1）の計算フローとプログラム，テストデータ，出力例を示す。なお，ここではサブルーチン "GESTO" のプログラムリストは載せていない（補遺3参照のこと）。

f_c 探索手法 (その1) のフロー



f. 探索手法 (その1) プログラムリスト

```

10 ' SAVE "10:KINEFRC",A
20 '
30 DIM X1(2),X2(2),R(50),RCQ(50),CCQ(50)
40 NCASE=10
50 P1=.6:P2=.4648
60 AA=8.9:'AREA KM^2
70 NR=15:NQ=31
80 FC=1.27 :DFC=.01
90 ' R DATA
100 FOR I=1 TO NR:READ R(I):NEXT I
110 DATA 0.5940,0.7920,0.6930,0.3960,0.4950,0.9900,0.6930,0.3960,1.1880
120 DATA 1.9799,3.0689,4.0588,3.0689,1.089,0.693
130 '
140 FOR I=1 TO NQ:READ RCQ(I):NEXT I
150 DATA 0.0422,0.0844,0.1766,0.1988,0.1609,0.1531,0.2753,0.2875,0.2797,0.4619
160 DATA 1.0841,1.6963,2.5784,2.7106,2.1828,1.6650,1.3972,1.1394,0.9116,0.6838
170 DATA 0.5159,0.3581,0.2503,0.2225,0.1747,0.1469,0.1191,0.0813,0.0734,0.0456
180 DATA 0.0378
190 FOR I=1 TO NR:SS=SS+R(I):NEXT I
200 RB=SS/NR:PRINT "RB=";RB
210 '
220 T=1/5:' CALCULATION STEP 1/5(HOUR)
230 T2=T*T:T3=T2*T:T4=T3*T
240 NL=INT(1/T+1E-03)
250 FCMIN=9999:JMIN=9999
260 FOR KK=1 TO NCASE
280 X(1)=0:X(2)=0:ER=0
290 FC=FC+DFC
300 K1=2.8235*FC*AA^.24
310 K2=.2835*K1^2*RB^-.2648:PRINT "K1,K2=";K1;K2
320 P3=P1/P2:K3=K1/K2
330 FOR J=1 TO NQ
340 Y1=X(1) :Y2=X(2)
350 FOR I=1 TO NL
360 GOSUB 1000:'SUBROUTINE GESTO
370 NEXT I
380 CCQ(J)=X(1)^(1/P2)
390 E=RCQ(J)-CCQ(J)
400 ER=ER+E^2
420 NEXT J
430 J=ER
440 IF J<JMIN THEN JMIN=J:FCMIN=FC:'PRINT "JMIN,FCMIN=";JMIN;FCMIN
450 LPRINT USING "KK=## FC=##.#### DFC=##.#### K1=##.#### K2=##.####
      J=##.####";KK;FC;DFC;K1;K2;J
460 NEXT KK
470 '
480 FC=FCMIN
490 K1=2.8235*FC*AA^.24 485 X(1)=0:X(2)=0:ER=0
500 K2=.2835*K1^2*RB^-.2648
510 P3=P1/P2:K3=K1/K2
520 FOR J=1 TO NQ
530 Y1=X(1) :Y2=X(2)
540 FOR I=1 TO NL
550 GOSUB 1000:'SUBROUTINE GESTO
560 NEXT I
570 CCQ(J)=X(1)^(1/P2)
580 E=RCQ(J)-CCQ(J)
590 ER=ER+E^2
600 NEXT J
610 J=SQR(ER/NQ)
620 GOSUB 640
630 END
640 ' OPTIMIZATION FC OUTPUT
650 LPRINT
660 LPRINT USING "NL=##";NL
670 LPRINT USING "NR=## NQ=##";NR;NQ
680 LPRINT USING "A=##.### RAVE=";AA;RB
690 LPRINT USING "FC=##.####";FCMIN
700 LPRINT USING "K1=##.#### K2=##.####";K1;K2
710 LPRINT
720 LPRINT "NO YR RCQ CCQ"
730 LPRINT " (MM/H) (MM/H) (MM/H)"
740 FOR J=1 TO NQ
750 LPRINT USING "## ##.### ##.### ##.###";J;R(J);RCQ(J);CCQ(J)
760 NEXT J
770 RETURN

```

テストデータ及び出力例

| | | | | | |
|--------------|-------------------|---------------------|-------------------|-------------------|--------------------|
| KK= 1 | FC= 1.2800 | DFC= 0.01000 | K1= 6.1073 | K2= 9.7736 | J= 0.4714 |
| KK= 2 | FC= 1.2900 | DFC= 0.01000 | K1= 6.1550 | K2= 9.9269 | J= 0.4588 |
| KK= 3 | FC= 1.3000 | DFC= 0.01000 | K1= 6.2028 | K2=10.0814 | J= 0.4501 |
| KK= 4 | FC= 1.3100 | DFC= 0.01000 | K1= 6.2505 | K2=10.2371 | J= 0.4443 |
| KK= 5 | FC= 1.3200 | DFC= 0.01000 | K1= 6.2982 | K2=10.3940 | J= 0.4414 |
| KK= 6 | FC= 1.3300 | DFC= 0.01000 | K1= 6.3459 | K2=10.5520 | J= 0.4413 ※ |
| KK= 7 | FC= 1.3400 | DFC= 0.01000 | K1= 6.3936 | K2=10.7113 | J= 0.4439 |
| KK= 8 | FC= 1.3500 | DFC= 0.01000 | K1= 6.4413 | K2=10.8718 | J= 0.4494 |
| KK= 9 | FC= 1.3600 | DFC= 0.01000 | K1= 6.4890 | K2=11.0334 | J= 0.4575 |
| KK=10 | FC= 1.3700 | DFC= 0.01000 | K1= 6.5368 | K2=11.1963 | J= 0.4683 |

※ 最小値

NL= 5
 NR=15 NQ=31
 A= 8.900 RAVE= 1.346
 FC= 1.3300
 K1= 6.3459 K2= 10.5520

NL : 1時間の分割数
 NR : 雨量データ個数
 NQ : 流量データ個数
 A : 流域面積 (km²)
 RAVE : 平均雨量強度 \bar{r} (mm/hr)
 FC : f_c 値
 K1, K2 : パラメータ k_1, k_2
 YR : 雨量 (有効雨量) (mm/hr)
 RCQ : 流量 (実測直接流出高) (mm/hr)
 CCQ : 流量 (計算直接流出高) (mm/hr)

| NO | YR (MM/H) | RCQ (MM/H) | CCQ (MM/H) |
|----|--------------|---------------|---------------|
| 1 | 0.594 | 0.042 | 0.077 |
| 2 | 0.792 | 0.084 | 0.107 |
| 3 | 0.693 | 0.177 | 0.160 |
| 4 | 0.396 | 0.199 | 0.218 |
| 5 | 0.495 | 0.161 | 0.270 |
| 6 | 0.990 | 0.153 | 0.340 |
| 7 | 0.693 | 0.275 | 0.422 |
| 8 | 0.396 | 0.288 | 0.477 |
| 9 | 1.188 | 0.280 | 0.543 |
| 10 | 1.980 | 0.462 | 0.099 |
| 11 | 3.069 | 1.084 | 1.034 |
| 12 | 4.059 | 1.696 | 1.643 |
| 13 | 3.069 | 2.578 | 2.329 |
| 14 | 1.089 | 2.711 | 2.568 |
| 15 | 0.693 | 2.183 | 2.305 |
| 16 | 0.000 | 1.665 | 1.811 |
| 17 | 0.000 | 1.397 | 1.316 |
| 18 | 0.000 | 1.139 | 0.938 |
| 19 | 0.000 | 0.912 | 0.674 |
| 20 | 0.000 | 0.684 | 0.495 |
| 21 | 0.000 | 0.516 | 0.374 |
| 22 | 0.000 | 0.358 | 0.290 |
| 23 | 0.000 | 0.250 | 0.231 |
| 24 | 0.000 | 0.223 | 0.188 |
| 25 | 0.000 | 0.175 | 0.157 |
| 26 | 0.000 | 0.147 | 0.132 |
| 27 | 0.000 | 0.119 | 0.113 |
| 28 | 0.000 | 0.081 | 0.098 |
| 29 | 0.000 | 0.073 | 0.086 |
| 30 | 0.000 | 0.046 | 0.076 |
| 31 | 0.000 | 0.038 | 0.068 |




図-4.4 最適 f_c を用いた計算ハイドログラフ

f_c 探索手法（その2）の説明

f_c 探索手法（その1）において問題となるのは Δf_c 値のオーダーである。 Δf_c 値を小さくすれば f_c 値は精度の高い値が得られるが計算時間がかかりすぎる。このため図-4・2に示された評価関数 J のカーブに着目し、ニュートン法（Newton-Raphson 法）を用いて f_c 値を同定することを試みる。

図-4・2に示されるように、ある洪水例に着目して f_c 値を探索すると、誤差関数 $J(f_c)$ は、最小値をただ1つだけ持つことがわかる。今、実測値と計算値の誤差 $e_i(f_c)$ および目的関数 $J(f_c)$ を次のように定義する。

$$e_i(f_c) = q_i^* - q_i(f_c) \quad \dots\dots\dots (8)$$

$$J(f_c) = \sum_{j=1}^{NQ} \{e_j(f_c)\}^2 \quad \dots\dots\dots (9)$$

次に、 f_c の微小変化 Δf_c を考えたとき、それに伴う $q_i(f_c)$ の変化を考えてみる。

$$\frac{\partial q_i(f_c)}{\partial f_c} \text{ (感度係数)} \quad \dots\dots\dots (10)$$

(10)式は、パラメータの変化に対する計算値の変化量を表しており、一般的に感度係数と呼ばれる。

今、 $f_c + \Delta f_c$ のときの誤差 $e_i(f_c + \Delta f_c)$ を考え、1次の項までテイラー級数展開すると、

$$e_i(f_c + \Delta f_c) = e_i(f_c) + \left(\frac{\partial e_i(f_c)}{\partial f_c} \right) \Delta f_c \quad \dots\dots\dots (11)$$

$$= e_i(f_c) - \frac{\partial q_i(f_c)}{\partial f_c} \Delta f_c \quad \dots\dots\dots (12)$$

よって、目的関数 $J(f_c + \Delta f_c)$ は次のように表わされる。

$$J(f_c + \Delta f_c) = \sum_{j=1}^{NQ} e_j^2(f_c + \Delta f_c)$$

あるいは(12)式を用いて、

$$J(f_c + \Delta f_c) = \sum_{j=1}^{NQ} \left(e_j(f_c) - \frac{\partial q_j(f_c)}{\partial f_c} \Delta f_c \right)^2 \quad \dots\dots\dots (13)$$




図-4・5 ニュートン法による最適値探索法

図-4・5に示すように、 J の最小化条件は $\partial J / \partial f_c = 0$ となった時点であり、 f_c 値の最適値が決定される。したがって、(13)式の両辺を Δf_c で微分してみる。

$$\frac{\partial J}{\partial \Delta f_c} = 2 \sum_{j=1}^{NQ} \left[\left\{ e_j(f_c) - \frac{\partial q_j(f_c)}{\partial f_c} \Delta f_c \right\} \left\{ - \frac{\partial q_j(f_c)}{\partial f_c} \right\} \right] \quad \dots\dots\dots (14)$$

(14)式の値が0となる点が J 値の最小値であることは前述した。よって、(14)式を0とおいて Δf_c について解くと

$$\Delta f_c = \frac{\sum_{j=1}^{NQ} \left\{ e_j(f_c) \times \frac{\partial q_j(f_c)}{\partial f_c} \right\}}{\sum_{j=1}^{NQ} \left\{ \frac{\partial q_j(f_c)}{\partial f_c} \right\}^2} \dots\dots\dots (15)$$

を得る。したがって、(15)式を用いれば、次回の f_c 値の更新量 Δf_c が逐次計算される。 f_c 値の収束条件は次式による。

$$\left| \frac{\Delta f_c}{f_c} \right| < \varepsilon \quad (\varepsilon = 0.001) \dots\dots\dots (16)$$

(15)式の算定上、感度係数 $\partial q_j / \partial f_c$ を求めておく必要がある。

今、(15)式で計算された Δf_c 値を m ステップ目で計算された値 Δf_c^m とすると、(10)式の感度係数は次の差分式で計算される。(図-4・6参照)

$$\frac{\partial q_j(f_c)}{\partial f_c} = \frac{q_j^m(f_c^m) - q_j^{m-1}(f_c^{m-1})}{\Delta f_c^{m-1}} \dots\dots (17)$$

ここで、

- $q_j^m(f_c^m)$: m ステップ目の計算流出高 (mm/hr),
- $q_j^{m-1}(f_c^{m-1})$: $m-1$ ステップ目の計算流出高 (mm/hr),
- Δf_c^{m-1} : $m-1$ ステップ目で計算された Δf_c 値。

次に f_c 値探索手法 (その2) の計算フロー、プログラム、テストデータ及び出力例を示す。P 158 の出力例を見ると、 f_c の補正值である Δf_c (DFC) が徐々に減少していくのがわかる。ここではサブルーチン "GESTO" のプログラムリストは載せていない (補遺3参照のこと)。





図-4・6 f_c 値と Δf_c 値の関係

f. 探索手法 (その2) のフロー



f. 探索手法 (その2) プログラムリスト

```

10 ' SAVE "10:KINEMATH",A
20 '
30 DIM X1(2),X2(2),R(50),RCQ(50),CCQ(50),BCQ(50),E(50)
40 NCASE=20
50 P1=.6:P2=.4648
60 AA=8.9:'AREA KM^2
70 NR=15:NQ=31
80 FC=.4 :DFC=.1
90 ' R DATA
100 FOR I=1 TO NR:READ R(I):NEXT I
110 DATA 0.5940,0.7920,0.6930,0.3960,0.4950,0.9900,0.6930,0.3960,1.1880
120 DATA 1.9799,3.0689,4.0588,3.0689,1.089,0.693
130 '
140 FOR I=1 TO NR:SS=SS+R(I):NEXT I
150 RB=SS/NR:PRINT "RB=";RB
160 '
170 ' Q DATA
180 FOR I=1 TO NQ:READ RCQ(I):NEXT I
190 DATA 0.0422,0.0844,0.1766,0.1988,0.1609,0.1531,0.2753,0.2875,0.2797,0.4619
200 DATA 1.0841,1.6963,2.5784,2.7106,2.1828,1.6650,1.3972,1.1394,0.9116,0.6838
210 DATA 0.5159,0.3581,0.2503,0.2225,0.1747,0.1469,0.1191,0.0813,0.0734,0.0456
220 DATA 0.0378
230 '
240 T=1/5:' CULCULATION STEP 1/5(HOUR)
250 T2=T*T:T3=T2*T:T4=T3*T
260 NL=INT(1/T+1E-03)
270 FOR KK=1 TO NCASE
280 SE1=0:SE2=0:ER=0
290 X(1)=0:X(2)=0
300 FC=FC+DFC
310 K1=2.8235*FC*AA^.24
320 K2=.2835*K1^2*RB^-.2648
330 P3=P1/P2:K3=K1/K2
340 FOR J=1 TO NQ
350 Y1=X(1) :Y2=X(2)
360 FOR I=1 TO NL
370 GOSUB 1000:'SUBROUTINE GESTO
380 NEXT I
390 CCQ(J)=X(1)^(1/P2)
400 E(J)=RCQ(J)-CCQ(J)
410 ER=ER+E(J)^2
420 IF KK=1 THEN 450
430 SE1=SE1+E(J)*(CCQ(J)-BCQ(J))/DFC
440 SE2=SE2+((CCQ(J)-BCQ(J))/DFC)^2
450 BCQ(J)=CCQ(J)
460 NEXT J
470 IF KK=1 THEN 520
480 J=ER
490 DFC=SE1/SE2
500 -LPRINT USING "KK=## FC=##.#### DFC=##.##### K1=##.#### K2=##.####
      J=##.####";KK;FC;DFC;K1;K2;J
510 IF ABS(DFC/FC)<1E-03 THEN 530
520 NEXT KK
530 ' OPTIMIZATION FC OUTPUT
540 LPRINT
550 LPRINT USING "NL=##";NL
560 LPRINT USING "NR=## NQ=##";NR;NQ
570 LPRINT USING "A=##.### RAVE=";AA;RB
580 LPRINT USING "FC=##.####";FC
590 LPRINT USING "K1=##.#### K2=##.####";K1;K2
600 LPRINT
610 LPRINT "NO YR RCQ CCQ"
620 LPRINT " (MM/H) (MM/H) (MM/H)"
630 FOR J=1 TO NQ
640 LPRINT USING "## ##.### ##.### ##.###";J;R(J);RCQ(J);CCQ(J)
650 NEXT J
660 END

```

テストデータ及び出力例

| | | | | | |
|-------|------------|--------------|------------|------------|-----------|
| KK= 2 | FC= 0.6000 | DFC= 0.39609 | K1= 2.8628 | K2= 2.1475 | J=10.7111 |
| KK= 3 | FC= 0.9961 | DFC= 0.20325 | K1= 4.7527 | K2= 5.9188 | J= 2.2267 |
| KK= 4 | FC= 1.1993 | DFC= 0.09990 | K1= 5.7225 | K2= 8.5807 | J= 0.6802 |
| KK= 5 | FC= 1.2992 | DFC= 0.02159 | K1= 6.1991 | K2=10.0696 | J= 0.4517 |
| KK= 6 | FC= 1.3208 | DFC= 0.00436 | K1= 6.3021 | K2=10.4070 | J= 0.4420 |
| KK= 7 | FC= 1.3252 | DFC= 0.00073 | K1= 6.3229 | K2=10.4758 | J= 0.4416 |

NL= 5
 NR=15 NQ=31
 A= 8.900 RAVE= 1.3464
 FC= 1.3252
 K1= 6.3229 K2= 10.4758

| NO | YR (MM/H) | RCQ (MM/H) | CCQ (MM/H) |
|----|--------------|---------------|---------------|
| 1 | 0.594 | 0.042 | 0.000 |
| 2 | 0.792 | 0.084 | 0.008 |
| 3 | 0.693 | 0.177 | 0.036 |
| 4 | 0.396 | 0.199 | 0.085 |
| 5 | 0.495 | 0.161 | 0.142 |
| 6 | 0.990 | 0.153 | 0.220 |
| 7 | 0.693 | 0.275 | 0.316 |
| 8 | 0.396 | 0.288 | 0.392 |
| 9 | 1.188 | 0.280 | 0.476 |
| 10 | 1.980 | 0.462 | 0.646 |
| 11 | 3.069 | 1.084 | 0.993 |
| 12 | 4.059 | 1.696 | 1.616 |
| 13 | 3.069 | 2.578 | 2.318 |
| 14 | 1.089 | 2.711 | 2.568 |
| 15 | 0.693 | 2.183 | 2.306 |
| 16 | 0.000 | 1.665 | 1.810 |
| 17 | 0.000 | 1.397 | 1.314 |
| 18 | 0.000 | 1.139 | 0.935 |
| 19 | 0.000 | 0.912 | 0.671 |
| 20 | 0.000 | 0.684 | 0.493 |
| 21 | 0.000 | 0.516 | 0.372 |
| 22 | 0.000 | 0.358 | 0.289 |
| 23 | 0.000 | 0.250 | 0.230 |
| 24 | 0.000 | 0.223 | 0.187 |
| 25 | 0.000 | 0.175 | 0.156 |
| 26 | 0.000 | 0.147 | 0.131 |
| 27 | 0.000 | 0.119 | 0.113 |
| 28 | 0.000 | 0.081 | 0.098 |
| 29 | 0.000 | 0.073 | 0.086 |
| 30 | 0.000 | 0.046 | 0.076 |
| 31 | 0.000 | 0.038 | 0.067 |

NL : 1時間の分割数
 NR : 雨量データ個数
 NQ : 流量データ個数
 A : 流域面積 (km²)
 RAVE : 平均雨量強度 \bar{r} (mm/hr)
 FC : f_c 値
 K1, K2 : パラメータ k_1, k_2
 YR : 雨量 (有効雨量) (mm/hr)
 RCQ : 流量 (実測直接流出高) (mm/hr)
 CCQ : 流量 (計算直接流出高) (mm/hr)




図-4.7 最適値 f_c を用いた計算ハイドログラフ